Recently, I found the following problem on Youtube:

$$x+y+z=1, \\ x^2+y^2+z^2=2, \\ x^3+y^3+z^3=3, \\ x^n+y^n+z^n= \,?$$

I did not continue to watch the video. So I have no idea what solution is presented. I decided to treat that as a challenge and to observe my steps, including the failed ones.

The problem sounds like a normal set of three equations with three unknowns that we can solve by eliminating. After all, this is the standard technique that we learn in school. But it is actually not that easy. Eliminating x from the second equation and solving for y yields an expression in y which contains a square root. Inserting everything in the third equation results in a mess.

Let us try numerically. We use Euler Math Toolbox here.

>function f2(x,y) &= 2-x^2-y^2-(1-x-y)^2

2                2    2
- y  - (- y - x + 1)  - x  + 2

>a3 &:= 3
3
>function f3(x,y) &= a3-x^3-y^3-(1-x-y)^3

3                3    3
- y  - (- y - x + 1)  - x  + 3

>plot2d("f2",levels=0,r=2);


We see, that the equations do not have a real solution. But they have, of course, a complex solution by the main theorem of Algebra.

The second idea is to try some tricks on the equations. E.g., we could multiply the first equation with x, y, and z and add the three results. Because of the first and the second equation, we get

$$xy+yz+xz=-\dfrac{1}{2}$$

That is a new equation. But is not clear how this is going to help.

After a while, the mathematical instinct kicks in to try something more generic and simple. Pattern matching inside the brain brings a known trick to the mind. Any sequence

$$1,x,x^2,x^3,\ldots$$

has an interesting property. If x is the zero of a polynomial

$$p(t)=t^3+at^2+bt+c$$

It satisfies the interesting property

$$x^n+ax^{n-1}+bx^{n-2}+cx^{n-3}=0$$

Thus the sequence above satisfies a recursion formula. Now, if x,y,z are the three zeros of the polynomial p(t) all three satisfy the same recursion formula, and so does the sum of the three zeros.

$$s_n = – (as_{n-1}+bs_{n-2}+cs_{n-3})$$

for

$$s_n = x^n+y^n+z^n$$

So we know from our three equations and the trivial equation

$$s_0 = x^0+y^0+z^0=3$$

that

$$3 = -(2a+b+3c)$$

But we also know that

$$p(t) = (t-x)(t-y)(t-z) = t^3 – (x+y+z) t^2 + (xy+yz+xz)t -xyz$$

Once we see that, we get a=-1 immediately from our problem. Having played around with tricks, we also know b=-1/2. From that, we now have

$$a=-1, \quad b=-\dfrac{1}{2}, \quad c=-\dfrac{1}{6}$$

We can verify that with Euler Math Toolbox.

For that, we compute the zeros of the polynomial and check the equations numerically. As expected, two of the zeros are complex.

>s=polysolve([-1/6,-1/2,-1,1])
[ -0.215425+0.264713i,  -0.215425-0.264713i,  1.43085+0i  ]
>sum(s)
1+0i
>sum(s^2)
2+0i
>sum(s^3)
3+0i


The expression for the real zero of p(t) is terrible and involves a third root.

$$z=u^{1/3} + \dfrac{5}{u^{1/3}} + \dfrac{1}{3}, \quad u=\dfrac{\sqrt{13}}{9 \cdot 2^{2/3}}+\dfrac{11}{54}$$

Thus, we get a recursion for the sum and thus for the equation sign in our problem.

$$s_n = s_{n-1}+\dfrac{1}{2}s_{n-2}+\dfrac{1}{6}s_{n-3}$$

Check with EMT.

>fraction sequence("x[-1]+x[-2]/2+x[-3]/6",[3,1,2,3],10)
[3,  1,  2,  3,  25/6,  6,  103/12,  221/18,  1265/72,  905/36]
>fraction real(sum(s^[0:9]')')
[3,  1,  2,  3,  25/6,  6,  103/12,  221/18,  1265/72,  905/36]


Can we make a problem with an easier solution? Sure! We only have to start with simpler zeros.

$$x+y+z=0, \\ x^2+y^2+z^2=14, \\ x^3+y^3+z^3=-18, \\ x^n+y^n+z^n= \,?$$

$$s_n = 1 + 2^n + (-3)^n$$

This kind of mathematical thinking looks like a touch of genius. But it is not. Not at all. It is just the summary of long experience in mathematics. The tricks involved are well known. They are just combined in the right way. Mathematical problem solving is just pattern matching using the patterns that have been exercised before. Maybe some stubbornness is needed, and a lot of trial and error. But no genius.

It does help very much to have a numerical and symbolic program like EMT available to check the computations.

## Ein Gedanke zu „Another Youtube Problem“

1. Lorenz Jaeneke

Very nice solution yielding a recursion. Here is an other aproach using simple linear Algebra:
When I name the tree given equations as e1, e2 and e3, we build four possible multiplicative combinations of them to contain the 5th power of x, y and z as linear terms namely e1^5, e1^3*e2, e1^3*e3 and e2*e3. In expended form they contain 21 different power products of some or all of the three variables. In mind we can assign to these products 21 new (but dependent) variables. So we have four new linear equations with 21 variables. Using the Gauss-Jordan-Normalization we receive as first line (first new equation): x^4+y^4+z^4=6. That’s all for n=6 and can be continued for higher values of n.

Diese Website verwendet Akismet, um Spam zu reduzieren. Erfahre mehr darüber, wie deine Kommentardaten verarbeitet werden.